
Hana Rudová

12 Rôzne 24

Podekovanie

Čast’ I

Zadania úloh

1 Unifikácia

1. Pre každú z nasledujúcich unifikácíı rozhodnite, či uspeje. Ak áno, zaṕı̌ste výslednú sub-
stitúciu

(a) pondelok = pondelok

(b) ’Pondelok’ = pondelok

(c) ’pondelok’ = pondelok

(d) Pondelok = pondelok

(e) pondelok = utorok

(f) den(pondelok) = pondelok

(g) den(pondelok) = X

(h) den(X) = den(pondelok)

(i) den(pondelok,X) = den(Y,utorok)

(j) den(pondelok,X,streda) = den(Y,utorok,X)

(k) den(pondelok,X,streda) = den(Y,stvrtok)

(l) den(D) = D

(m) vikend(den(sobota),den(nedela)) = vikend(X,Y)

(n) vikend(den(sobota),X) = vikend(X,den(nedela))

2 Backtracking, rez

Fibonacciho postupnost’ je postupnost’ č́ısel, ktorej prvé dva členy sú rovné 1 a každý d’aľśı člen je
súčtom dvoch predchádzajúcich (fib(1) = 1, fib(2) = 1, fib(n) = fib(n− 1) + fib(n− 2), n > 2).

1. Nižšie uvedený logický program obsahuje predikát fib/2, ktorý vypoč́ıta n-tý člen Fibonac-
ciho postupnosti. Program však nie je úplne korektný. Správnu hodnotu śıce vypoč́ıta, no
ak si po jej nájdeńı vyžiadame pomocou ; d’aľsie riešenia, program sa zacykĺı (správne by
mal skončit’ s odpoved’ou no). Pridajte do programu na správne miesta operátory rezu tak,
aby program v takýchto pŕıpadoch necyklil.

fib(N,1) :- N =< 2.

fib(N,X) :- N1 is N-2, fib(N1,1,1,X).

fib(0,_A,X,X).

fib(N,A,B,X) :- N1 is N-1, C is A+B, fib(N1,B,C,X).

2. Zamyslite sa a zdôvodnite, prečo nasledujúci logický program, ktorý poč́ıta n-tý člen Fi-
bonacciho postupnosti priamo podl’a defińıcie, nie je vhodný (napriek tomu, že výsledok
vypoč́ıta vždy korektne)

fib(1,1).

fib(2,1).

fib(N,X) :- N>2, N1 is N-1, N2 is N-2, fib(N1,Y),fib(N2,Z),X is Y+Z.

3. Naṕı̌ste predikát convert/2, ktorý prevedie svoj prvý argument (prirodzené č́ıslo) na repre-
zentáciu daného č́ısla pomocou následńıkov. Napr.:
?-convert(3,X).

X = s(s(s(0)))

4. Naṕı̌ste predikát geq/2, ktorý ako argumenty dostane 2 č́ısla zadané pomocou následńıkov
a uspeje, ak prvé č́ıslo je väčšie alebo rovné druhému. Napr.:
?-geq(s(0),s(s(0))).

no

5. Naṕı̌ste predikát sucet/3, ktorý vypoč́ıta súčet dvoch č́ısel zadaných pomocou následńıkov.
Napr.:
?- sucet(s(0),s(s(0)),X).

X = s(s(s(0)))

6. Naṕı̌ste predikát sucin/3, ktorý vypoč́ıta súčin dvoch č́ısel zadaných pomocou následńıkov.
Napr.:
?- sucin(s(s(0)),s(s(0)),X).

X = s(s(s(s(0))))

7. Zadaná je databáza faktov v tvare priamacesta(a,b), ktoré hovoria o tom, že existuje
priama (jednosmerná) cesta z miesta a do miesta b. Naṕı̌ste predikát cesta/2 tak, že
cesta(a,b). uspeje, ak existuje nejaká cesta (ktorá môže prechádzat’ cez l’ubovol’ný počet
miest) z miesta a do miesta b. Môžete predpokladat’, že cesty netvoria cykly (tzn. ak raz
od́ıdete z nejakého miesta, už sa tam nedá vrátit’)
Rozš́ırenie 1: Upravte vaše riešenie tak, že cesta(a,b,S). uspeje, ak existuje nejaká cesta a
S je zoznam miest, cez ktoré táto cesta prechádza (ideálne v porad́ı, v akom ich prechádzame)
Rozš́ırenie 2: Upravte vaše riešenie tak, aby fungovalo aj v pŕıpade, že cesty tvoria cykly.

3 Aritmetika

1. Rozhodnite, ktoré z nasledujúcich dotazov uspejú a ktoré nie:

15 is 3*5.

14 =\= 3*5.

15 = 3*5.

15 == 3*5.

15 =\= 3*5.

15 =:= 3*5.

3*5 == 3*5.

3*5 == 5*3.

3*5 =:= 3*5.

10-3 =\= 9-2.

2. Naṕı̌ste predikát mocnina/1, ktorý uspeje ak č́ıslo zadané ako argument je mocninou dvojky
(využite fakt, že mocniny dvojky je možné opakovane bez zvyšku delit’ dvomi až kým nedos-
taneme jednotku)

3. Naṕı̌ste predikát sucet/2, ktorý spoč́ıta súčet prvých N prirodzených č́ısel. Napr.:
?- sucet(5,X).

X = 15

4. Naṕı̌ste predikát nsd/3, ktorý vypoč́ıta najväčšieho spoločného delitel’a dvoch č́ısel. Napr.:
?- nsd(21,49,X).

X = 7

Využite Euklidov algoritmus, ktorý je založený na pozorovańı, že najväčš́ı spoločný delitel’

č́ısel a, b kde a > b je rovnaký ako najväčš́ı spoločný delitel’ č́ısel (a− b), b

5. Naṕı̌ste predikát cifsucet/2, ktorý spoč́ıta ciferný súčet zadaného č́ısla. Napr.:
?- cifsucet(12345,X).

X = 15

4 Zoznamy

1. Ktoré z nasledujúcich zápisov predstavujú korektný zápis zoznamu? Aký počet prvkov majú
v týchto pŕıpadoch pŕıslušné zoznamy?

[1|[2,3,4]]

[1,2,3|[]]

[1|2,3,4]

[1|[2|[3|[4]]]]

[1,2,3,4|[]]

[[]|[]]

[[1,2]|4]

[[1,2],[3,4]|[5,6,7]]

2. Naṕı̌ste predikát dvojnasobok/2, ktorý uspeje ak prvky v druhom zozname sú dvojnásobkami
prvkov prvého zoznamu.
napr. dvojnasobok([5,3,2],[10,6,4]) uspeje, ale dvojnasobok([1,2,3],[2,4,5]) ne-
uspeje

3. Naṕı̌ste predikát filter/2, ktorý zo zoznamu odstráni všetky neč́ıselné hodnoty, napr.:
?- filter([5,s(0),3,a,2,A,7],X).

X = [5,3,2,7]

4. Naṕı̌ste predikát cifry/2, ktorý zo zoznamu cifier vybuduje č́ıslo, napr.:
?- cifry([2,8,0,7],X).

X = 2807

5. Naṕı̌ste predikát nty/3, ktorý vráti n-tý prvok zoznamu, napr.:
?- nty(4,[5,2,7,8,0],N).

N = 8

6. Napšte predikát rovnaju sa/2, ktorý uspeje, ak sa dva zadané zoznamy rovnajú. V pŕıpade,
že sa zoznamy nerovnajú, vyṕı̌ste dôvod, prečo je to tak (jeden zoznam je kratš́ı ako druhý,
výpis prvých 2 prvkov, ktoré sa nerovnajú...) Môžete predpokladat’, že zoznamy neobsahujú
premenné. Napŕıklad:
?- rovnaju sa([5,3,7],[5,3,7]).

yes

?- rovnaju sa([5,3,7],[5,3,7,2]).

1. zoznam je kratsi

no

?- rovnaju sa([5,3,7],[5,2,3,7]).

3 sa nerovna 2

no

7. Naṕı̌ste predikát priemer/2, ktorý vypoč́ıta aritmetický priemer zadaného zoznamu. Napr.:
?- priemer([1,2,3,4,5,6],X).

X = 3.5

8. Naṕı̌ste predikát zip/3, ktorý z dvoch zoznamov vytvoŕı nový spojeńım pŕıslušných prvkov
do dvoj́ıc. Napr.:
zip([1,2,3],[4,5,6],S).

S = [1-4,2-5,3-6]

9. Naṕı̌ste predikát insert/3, ktorý do usporiadaného zoznamu č́ısel vlož́ı d’aľsie č́ıslo tak, aby
aj výsledný zoznam bol usporiadaný. Napr.:
?- insert(7,[1,2,3,10,12],X).

X = [1,2,3,7,10,12]

10. Naṕı̌ste predikát merge/3, ktorý spoj́ı 2 vzostupne usporiadané zoznamy č́ısel tak, aby
výsledný zoznam bol opät’ vzostupne usporiadaný.

5 Vstup a výstup

1. Naṕı̌ste predikát wordcount/2, ktorý ako prvý parameter dostane názov textového súboru
obsahujúci slová vo formáte slovo. a ako svoj druhý argument vráti zoznam slov a počet ich
výskytov v texte. Každé slovo sa má vo výslednom zozname nachádzat’ práve raz. Výsledný
zoznam teda môže vyzerat’ napŕıklad takto: [a-10,ahoj-4,...].

2. Naṕı̌ste predikát lettercount/2, ktorý ako prvý parameter dostane názov textového súboru
a ako svoj druhý argument vráti zoznam obsahujúci počet výskytov jednotlivých ṕısmen
abecedy (pre jednoduchost’ predpokladajme, že súbor obsahuje len malé ṕısmená). Os-
tatné znaky by mali byt’ ignorované. (formát výstupného zoznamu si zvol’te podl’a vlastného
uváženia)

6 Dekompoźıcia termu

1. Naṕı̌ste predikát map/3 tak, že zavolanie map(Funkcia,Z1,Z2) aplikuje funkciu Funkcia

postupne na všetky prvky zoznamu Z1 a výsledky vlož́ı do zoznamu Z2. Ak napŕıklad máme
dvojnasobok(X,Y) :- Y is 2*X. potom po zavolańı map(dvojnasobok,[5,3,2],S).
dostaneme S = [10,6,4]. Pri defińıcii využite konštrukciu pomocou =.. a predikát call/1.

2. Naṕı̌ste predikát nahrad/4 tak, že nahrad(T1,T2,T3,T4) (kde T1,T2,T3,T4 sú termy bez
premenných) uspeje práve ak T4 vznikol z T1 nahradeńım každého výskytu T2 termom T3.
Napŕıklad:
?- nahrad(a(a(a)),a,b,X).

X = a(a(b))

?- nahrad([sin(pi/4),sqrt(sin(pi/2)*2+4),2*sin(pi/2)],sin(pi/2),1,X).

X = [sin(pi/4),sqrt(1*2+4),2*1]

3. Naṕı̌ste predikát position(Subterm,Term,Position) kde Position je zoznam identifi-
kujúci prvú poźıciu termu Subterm v terme Term. Napŕıklad:
?- position(5,2*f(5),X).

X = [2,1]

pretože f(5) je druhý argument pre operátor * a 5 je prvý argument f

4. Naṕı̌ste predikát copy/2, tak, že copy(T1,T2) uspeje ak term T2 je kópiou termu T1.
Tzn. termy majú rovnakú štruktúru, ale použ́ıvajú rôzne premenné (f(X,g(X)) je kópiou

f(Y,g(Y)) ale nie je kópiou f(X,g(X)) ani f(U,g(V))). Snažte sa, aby sa chovanie predikátu
copy čo najviac podobalo na chovanie zabudovaného predikátu copy term/2.
Nápoveda: použite pomocný zoznam na zapamätanie si premenných v pôvodnom terme a v
kópii.

7 Všetky riešenia

1. Uvažujme nasledujúcu databázu faktov:

f(a,b).

f(a,c).

f(a,d).

f(e,c).

f(g,h).

f(g,b).

f(i,a).

Zistite bez použitia Prologu aká bude odpoved’ na dotaz:

(a) findall(X,f(a,X),List).

(b) findall(X,f(X,b),List).

(c) findall(X,f(X,Y),List).

(d) bagof(X,f(X,Y),List).

(e) setof(X,Y^f(X,Y),List).

2. Naṕı̌ste predikát subsets/2, ktorý pre danú množinu vygeneruje množinu všetkých jej
podmnož́ın (množinou rozumieme zoznam, v ktorom sa prvky neopakujú). Na porad́ı podmnož́ın
nezálež́ı. Môžete si skúsit’ naṕısat’ aj predikát isset/1 ktorý uspeje ak zadaný zoznam ko-
rektne reprezentuje množinu, tzn. neobsahuje duplicitné prvky.

3. Naṕı̌ste predikát permutations/2, ktorý pre daný zoznam vygeneruje zoznam všetkých jeho
permutácíı v lexikografickom porad́ı. Napr.:
:- permutations([1,2,3],S).

S = [[1,2,3],[1,3,2],[2,1,3],[2,3,1],[3,1,2],[3,2,1]]

Môžete použit’ zabudovaný predikát permutation/2 z library(lists)

4. Naṕı̌ste predikát variations(+S,+K,-Z) tak, že Z bude obsahovat’ všetky K-prvkové variácie
zoznamu S.

8 Rezolúcia

1. Nájdite (ak existuje) najobecneǰśı unifikátor σ nasledujúcich literálov:

(a) p(X) a p(f(Y ))

(b) p(X, f(X)) a p(f(X), Y )

(c) p(X,Y ) a p(Z,Z)

(d) p(X, 8) a p(Y, Y )

(e) p(f(X)) a p(Z, Y )

(f) p(f(Z), g(X)) a p(Y, g(Y ))

(g) p(X, g(Z), X) a p(f(Y ), Y,W )

2. Na literáloch C1, C2 vykonajte rezolúciu a nájdite rezolventu, jednotlivé kroky (premenovanie
premenných, nájdenie najobecneǰsieho unifikátora, rezolučný prinćıp) rozṕı̌ste:

(a) C1 = {q(X),¬r(Y ), p(f(Z), f(Z))} a C2 = {n(Y ),¬r(W ),¬p(f(W ), f(W ))}
(b) C1 = {q(X),¬r(Y ), p(X,Y )} a C2 = {n(Y ),¬r(W ),¬p(f(W ), f(W ))}

3. Pomocou lineárnej rezolúcie vyvrát’te

S = {{¬t(X)}, {s(1)}, {c(1), c(2),¬s(1)}, {t(0),¬c(1)}, {t(1),¬c(1)}, {t(0),¬c(2)}, {t(2),¬c(2)}}

4. Preṕı̌ste nasledujúci program v prologovskej notácii ako množinu klauzuĺı. Následne na cieli
?- a. demonštrujte lineárnu rezolúciu.

a :- b,c,d.

b :- f.

c.

d.

f.

5. Pomocou rezolučného prinćıpu vyvrát’te S = {{a, b}, {c,¬b}, {¬c}, {¬a, d}, {¬e}, {e,¬d}}.
Existuje viacero možných riešeńı, skúste nájst’ čo najviac z nich.

6. Uvážme nasledujúci Prologovský program a dotaz ?- a. Nakreslite zodpovedajúci SLD
strom.

a :- b,c.

a :- d.

b.

b :- d.

c.

d :- e.

d.

7. Uvážme nasledujúci Prologovský program a dotaz ?- p(X,X). Nakreslite zodpovedajúci
SLD strom.

p(X,Y) :- q(X,Z), r(Z,Y).

p(X,X) :- s(X).

q(X,b).

q(b,a).

q(X,a) :- r(a,X).

r(b,a).

s(X) :- t(X,a).

s(X) :- t(X,b).

s(X) :- t(X,X).

t(a,b).

t(b,a).

8. Do programu z predchádzajúcej úlohy pridajte 1 operátor rezu, tak aby výsledný program
uspel práve dvakrát (pomôžte si nakresleným stromom z predchádzajúcej úlohy).

9 Negácia v logickom programovańı

1. Rozhodnite, či nasledujúce programy sú stratifikované. Ak áno, nájdite nejakú ich strati-
fikáciu:

(a) p(X):-q(X), \+ r(X).

p(X):-q(X), \+ t(X).

r(X):-s(X), \+ t(X).

t(a).

s(a).

s(b).

q(a).

(b) p :- q, \+ r

r :- s, \+ p

q.

s.

(c) p :- \+ q, \+ r

q :- \+ s

r :- \+ t

s.

t.

2. Uvážme logický program:

f(X) :- \+ g(X).

g(X) :- \+ f(X).

g(5).

Pre ciele :- f(5) a :- ¬ f(5) rozhodnite, či existuje ich odvodenie

(a) podl’a closed world assumption

(b) podl’a negation as failure

3. Uvažujme nasledujúci logický program:

a(X) :- c(X), Y is X+1, \+ d(Y).

a(X) :- c(X), \+ d(X).

b(X) :- X > 0.

c(X) :- d(X), Y is X+1, \+ c(Y).

d(1).

d(2).

a dotaz :- a(P).

Naṕı̌ste množinu SLDNF odvodeńı pre tento dotaz a uved’te o aký typ odvodenia sa jedná.

10 Logické programovanie s obmedzujúcimi podmienkami

1. Naṕı̌ste predikát labeling/1 ktorý ako aktuálnu premennú vyberá premennú s najmenšou
vel’kost’ou domény a hodnoty prirad’uje vzostupne (pomocou indomain)

2. Zistite aké budú domény A,B,C po AC-3 pre domain([A,B,C],1,10), A#=2*B+1, C#<A

3. Ako sa zmenia domény z predchádzajúcej úlohy ak miesto AC-3 použijeme konzistenciu
medźı?

4. Ako vyzerá stavový priestor pre obmedzenia A,B,C in 1..4, A#=B+1, C#<B pri usporia-
dańı premenných A,B,C a použit́ı:

(a) Backtrackingu

(b) Kontroly dopredu (forward checking)

(c) Pohl’adu dopredu (looking ahead)

5. Vyriešte pomocou Prologu nasledujúci algebrogram:

KC + I = OK

+ + +

A + A = KM

----------------------------

OL + KO = LI

6. Naṕı̌ste program pre hl’adanie riešeńı ,,zmenšenej” verzie Sudoku (hracie pole má rozmery

4 × 4 a je rozdelené na 4 štvorce 2 × 2, v každom riadku, st́lpci a štvorci sa muśı každé z
č́ısel 1 až 4 nachádzat’ práve raz.). Ako rozš́ırenie môžete pridat’ aj pekne formátovaný výpis
nájdeného riešenia.

7. Vyriešte pomocou Prologu nasledujúcu variáciu (jednu z mnoha) známej Einsteinovej hádanky:
Je rada piatich domov, pričom každý má inú farbu. V týchto domoch žije pät’ l’ud́ı rôznych
národnost́ı. Každý z nich chová iné zviera, rád pije iný nápoj a fajč́ı iné cigarety.

I. Brit býva v červenom dome.

II. Švéd chová psa.

III. Dán pije čaj.

IV. Zelený dom stoj́ı hned’ vl’avo od bieleho.

V. Majitel’ zeleného domu pije kávu.

VI. Ten, kto fajč́ı Pall Mall, chová vtáka.

VII. Majitel’ žltého domu fajč́ı Dunhill.

VIII. Človek z prostredného domu pije mlieko.

IX. Nór býva v prvom dome.

X. Ten, kto fajč́ı Blend, býva vedl’a toho, kto chová mačku.

XI. Ten, kto chová kone, býva vedl’a toho, kto fajč́ı Dunhill.

XII. Ten, kto fajč́ı Blue Master, pije pivo.

XIII. Nemec fajč́ı Prince.

XIV. Nór býva vedl’a modrého domu.

XV. Ten, kto fajč́ı Blend, má suseda, ktorý pije vodu.

Kto chová rybičky?

Úloha má jediné riešenie, je možné jednoznačne určit’ všetky informácie, nielen informáciu o
tom, kto chová rybičky. Najzložiteǰsou čast’ou úlohy teda je zvolit’ si správnu reprezentáciu
zadaných faktov, následne je zvyšok relat́ıvne jednoduchý. Ak sa vám to nedaŕı, pri riešeńı
v závere zbierky nájdete nápovedu.

11 DC gramatiky

1. Naṕı̌ste DCG pre rozpoznávanie párnych (=sudých) binárnych č́ısel.

2. Naṕı̌ste DCG rozpoznávajúcu palindrómy pozostávajúce z ṕısmen a, b (palindróm je slovo,
ktoré sa č́ıta rovnako od začiatku aj od konca. Napr. [], [a,b,a], [a,b,b,a] sú pa-
lindrómy, ale [a,b,b] palindróm nie je)

3. Naṕı̌ste DCG pre jazyk, ktorý obsahuje slová v tvare anbm kde m ≥ n (pomôcka: vyskúšajte
najprv jazyk slov v tvare anbn)

4. Naṕı̌ste DCG pre jazyk slov v tvare anb2n.

5. Naṕı̌ste DCG pre jazyk slov v tvare anbn+1cn+2.

6. Preved’te nasledujúcu gramatiku na predikáty s rozdielovými klauzulami:

s --> np, vp.

np --> det, n.

vp --> tv, np.

vp --> v.

det --> [the].

det --> [a].

det --> [every].

n --> [man].

n --> [woman].

n --> [park].

tv --> [loves].

tv --> [likes].

v --> [walks].

7. Nakreslite derivačný strom pre nejakú vetu rozpoznávanú gramatikou z predchádzajúcej
úlohy.

8. Naṕı̌ste, aké termy zodpovedajú nasledujúcim derivačným stromom:

Hana Rudová, Fakulta informatiky, Masarykova univerzita 11

S

Ad

Yesterday

S1

NP

John

VP

V

saw

NP

Det

a

NP

Adj

good

NP

N

movie

Exp

Sub

Exp

Add

Exp

Num

5

Exp

Mul

Exp

Num

7

Exp

Num

2

Exp

Num

4

9. Pre prvý strom z predchádzajúcej úlohy zostrojte zodpovedajúcu gramatiku s vytvárańım
derivačného stromu (tzn. tak aby derivačný strom vety “Yesterday John saw a good movie”
zodpovedal znázornenému - overte si to v Prologu). Nevad́ı, ak vaša gramatika bude generovat’

aj iné, nezmyselné vety.

12 Rôzne

1. Zistite čo rob́ı nasledujúci zámerne škaredo naṕısaný predikát a preṕı̌ste ho tak, aby bol
zrozumitel’neǰśı (pri zachovańı logickej ekvivalencie s pôvodnou defińıciou). Pri zist’ovańı, čo
program rob́ı, využite trasovanie v Prologu (predikát trace/0).

foo(X) :- (\+ (X = 1)),(X=2;X=3;foo(X,X)).

foo(X,Y) :-

(Y=X -> (Z is X-1,foo(X,Z)); (Y=1 ; (P is (X // Y) * Y, \+(P=X), Z is Y-1, foo(X,Z)))).

Čast’ II

Riešenia úloh

1 Unifikácia

1. výsledky si l’ahko oveŕıte v prologu

2 Backtracking, rez

1. fib(N,1) :- N =< 2, !.

fib(N,X) :- N1 is N-2, fib(N1,1,1,X).

fib(0,_A,X,X) :- !.

fib(N,A,B,X) :- N1 is N-1, C is A+B, fib(N1,B,C,X).

2. Program je vel’mi neefekt́ıvny (až exponenciálna zložitost’), pretože hodnoty predchádzajúcich
členov postupnosti sa poč́ıtajú opakovane. Najlepšie to uvid́ıte, ak si nakresĺıte strom výpočtu
pre nejaké malé N, napr. 6. Takto zaṕısaný program navyše neumožňuje optimalizáciu po-
sledného volania.

3. convert(0,0).

convert(X,s(Y)) :- X1 is X-1, convert(X1,Y).

4. geq(_,0).

geq(s(X),s(Y)) :- geq(X,Y).

5. sucet(0,X,X).

sucet(s(X),Y,s(Z)):-sucet(X,Y,Z).

6. sucin(_X,0,0).

sucin(X,s(Y),Z) :- sucin(X,Y,Z1), sucet(X,Z1,Z).

7. Riešenie, ktoré zahŕňa obe rozš́ırenia:

(1) :- dynamic navstivene/1.

(2) cesta(A,A,[A]).

(3) cesta(A,B,[A|S]) :-

(4) assert(navstivene(A)),

(5) priamacesta(A,X),

(6) \+ navstivene(X),

(7) cesta(X,B,S).

(8) cesta(A,_B,_S) :- retract(navstivene(A)), fail.

Pre riešenie bez druhého rozš́ırenia stač́ı vynechat’ riadky 1,4,6,8. Alernat́ıvne by sme mohli
druhé rozš́ırenie riešit’ pridańım d’aľsieho argumentu - zoznamu už navšt́ıvených vrcholov.
Ak by sme nechceli riešit’ ani prvé rozš́ırenie, môžeme z predikátov odstránit’ tret́ı argument.
Stoj́ı za zmienku, že úloha bez rozš́ıreńı je v podstate ekvivalentá úlohe z cvičeńı s faktami
v tvare rodic(x,y). a s predikátom potomek/2.

3 Aritmetika

1. Výsledky si l’ahko oveŕıte v prologu.

2. mocnina(1).

mocnina(X) :- X mod 2 =:= 0, X1 is X // 2, mocnina(X1).

3. Funkčné riešenie:

sucet(N,S) :- sucet(N,0,S).

sucet(0,S,S) :- !.

sucet(N,X,S) :- N1 is N-1, X1 is X+N, sucet(N1,X1,S).

Lepšie riešenie (využ́ıva vzorec pre súčet aritmetickej postupnosti):

sucet(N,S) :- S is (N*(N+1)) // 2.

4. nsd(X,X,X).

nsd(A,B,X) :- A>B, A1 is A-B, nsd(A1,B,X).

nsd(A,B,X) :- B>A, B1 is B-A, nsd(A,B1,X).

Dodajme, že existuje efekt́ıvneǰsia verzia využ́ıvajúca zvyšky po deleńı.

5. cifsucet(N,CS) :- cifsucet(N,0,CS).

cifsucet(0,CS,CS) :- !.

cifsucet(N,X,CS) :- X1 is X + N mod 10, N1 is N //10, cifsucet(N1,X1,CS).

4 Zoznamy

1. Výsledky si l’ahko oveŕıte v Prologu, na zistenie počtu prvkov zoznamu môžete použit’

length/2

2. dvojnasobok([],[]).

dvojnasobok([X1|T1],[X2|T2]) :- X2 is 2*X1, dvojnasobok(T1,T2).

3. filter([],[]).

filter([X|T1],[X|T2]) :- number(X), !, filter(T1,T2).

filter([_X|T1],T2) :- filter(T1,T2).

4. cifry(S,X) :- cifry(S,0,X).

cifry([],X,X).

cifry([A|T],P,X) :- P1 is 10*P + A, cifry(T,P1,X).

5. nty(N,[X|_T],Y) :- N =< 1, !, X = Y.

nty(N,[_X|T],Nty) :- N1 is N-1, nty(N1,T,Nty).

6. rovnaju_sa([],[]).

rovnaju_sa([X|T1],[X|T2]) :- !, rovnaju_sa(T1,T2).

rovnaju_sa([],_S) :- write(’1. zoznam je kratsi’), !, fail.

rovnaju_sa(_S,[]) :- write(’2. zoznam je kratsi’), !, fail.

rovnaju_sa([X|_T1],[Y|_T2]) :- write(X), write(’ sa nerovna ’), write(Y), !, fail.

Pre úplnost’ dodajme, že ak by sme nepožadovali výpis dôvodu nerovnosti, vystačili by sme
si s faktom rovnaju sa(S,S). (ak S neobsahuje premenné)

7. priemer(S,X) :- priemer(S,0,0,X).

priemer([],Sum,Count,X) :- X is Sum/Count.

priemer([H|T],Sum,Count,X) :- Sum1 is Sum+H, Count1 is Count+1, priemer(T,Sum1,Count1,X).

8. zip([H1|T1],[H2|T2],[H1-H2|T]) :- zip(T1,T2,T), !.

zip(_,_,[]). %aspon 1 zoznam je prazdny

9. insert(X,[],[X]).

insert(X,[H|T],[X,H|T]) :- X =< H, !.

insert(X,[H|T],[H|T1]) :- insert(X,T,T1).

10. merge([],X,X) :- !. %bez rezu by merge([],[],S). uspel dvakrat

merge(X,[],X).

merge([H1|T1],[H2|T2],[H1|T]) :- H1 =< H2, !, merge(T1,[H2|T2],T).

merge([H1|T1],[H2|T2],[H2|T]) :- merge([H1|T1],T2,T).

5 Vstup a výstup

1. Jedno z možných riešeńı s využit́ım dynamickej modifikácie databázy:

:- dynamic slovo/2.

wordcount(File,S) :-

retractall(slovo(_,_)),

seeing(Current),

see(File),

repeat,

read(C),

(slovo(C,X) -> (Y is X+1, retract(slovo(C,X))); Y is 1),

assert(slovo(C,Y)),

C == end_of_file,

retract(slovo(end_of_file,1)),

!,

seen,

see(Current),

findall(Z-Count,slovo(Z,Count),S).

2. :- dynamic znak/2.

lettercount(File,S) :-

retractall(znak(_,_)),

seeing(Current),

see(File),

repeat,

get_char(C),

(znak(C,X) -> (Y is X+1, retract(znak(C,X))); Y is 1),

(member(C,[a,b,c,d,e,f,g,h,i,j,k,l,m,n,o,p,q,r,s,t,u,v,w,x,y,z])

-> assert(znak(C,Y));

true),

C == end_of_file,

!,

seen,

see(Current),

setof(Z-Count,znak(Z,Count),S).

6 Dekompoźıcia termu

1. map(_Funkcia, [], []).

map(Funkcia, [H1|T1], [H2|T2]) :-

F =.. [Funkcia, H1, H2], call(F), map(Funkcia, T1, T2).

2. nahrad(X,_,_,X) :- var(X),!.

nahrad(Co,Co,Cim,Cim) :- !.

nahrad(S,_,_,S) :- atomic(S), !.

nahrad([X|T],Co,Cim,[X1|T1]) :- nahrad(X,Co,Cim,X1),nahrad(T,Co,Cim,T1),!.

nahrad(X,Co,Cim,X1) :- compound(X), X =.. [H|T], nahrad(T,Co,Cim,T1), X1 =.. [H|T1].

3. position(S,S,[]) :- !.

position(S,[H|_],[1|T]) :- position(S,H,T),!.

position(S,[_|T],[X1|T1]) :- position(S,T,[X2|T1]),X1 is X2+1, !.

position(S,T,L) :- compound(T), T =..[_H|A], position(S,A,L).

4. copy(A,B,Z):-cp(A,[],B,Z).

cp(A,Vars,A,Vars):-

atomic(A).

cp(V,Vars,NV,NVars):-

var(V),register_var(V,Vars,NV,NVars).

cp(Term,Vars,NTerm,NVars):-

compound(Term),

Term=..[F|Args], % decompose term

cp_args(Args,Vars,NArgs,NVars),

NTerm=..[F|NArgs]. % construct copy term

cp_args([H|T],Vars,[NH|NT],NVars):-

cp(H,Vars,NH,SVars),

cp_args(T,SVars,NT,NVars).

cp_args([],Vars,[],Vars).

register_var(V,[X/H|T],N,[X/H|NT]):-

V\==X, % different variables

register_var(V,T,N,NT).

register_var(V,[X/H|T],H,[X/H|T]):-

V==X. % same variables

register_var(V,[],N,[V/N]).

7 Všetky riešenia

1. Výsledky si l’ahko oveŕıte v Prologu

2. subsets(X,S) :- isset(X), findall(Y,subset(X,Y),S).

isset([]).

isset([X|T]) :- \+ member(X,T), isset(T).

subset([],[]).

subset([_X|T],S) :- subset(T,S).

subset([X|T],[X|S]) :- subset(T,S).

3. :- use_module(library(lists)).

permutations(S,Z) :- setof(X, permutation(S,X), Z).

4. :- use_module(library(lists)).

variations(S,K,Z) :- findall(Y, (subset(S,X), length(X,K), permutation(X,Y)), Z)

8 Rezolúcia

1. (a) σ = [X/f(Y )]

(b) neexistuje

(c) σ = [X/Z, Y/Z]

(d) σ = [X/8, Y/8]

(e) neexistuje

(f) σ = [X/f(Z), Y/f(Z)]

(g) σ = [W/f(g(Z)), X/f(g(Z)), Y/g(Z)]

2. (a) Premenovanie premenných ρ = [Y/A]
C1 = {q(X),¬r(A), p(f(Z), f(Z))}, C2 = {n(Y ),¬r(W ),¬p(f(W ), f(W ))}
Najobecneǰśı unifikátor σ = [Z/W ]
C1 = {q(X),¬r(A), p(f(W ), f(W ))}, C2 = {n(Y ),¬r(W ),¬p(f(W ), f(W ))}
Z rezolučného prinćıpu dostávame C = {q(X),¬r(A), n(Y ),¬r(W )}

(b) Premenovanie premenných ρ = [Y/A]
C1 = {q(X),¬r(A), p(X,A)}, C2 = {n(Y ),¬r(W ),¬p(f(W ), f(W ))}
Najobecneǰśı unifikátor σ = [X/f(W ), A/f(W )]
C1 = {q(f(W )),¬r(f(W )), p(f(W ), f(W ))}, C2 = {n(Y ),¬r(W ),¬p(f(W ), f(W ))}
Z rezolučného prinćıpu {q(f(W )),¬r(f(W )), n(Y ),¬r(W )}

3. Jeden z možných postupov

�

{t(2)}

{c(2)}

{c(2),¬s(1)}

{¬c(1)}

{¬t(X)} {t(1),¬c(1)}

{c(1), c(2),¬s(1)}

{s(1)}

{t(2),¬c(2)}

{¬t(X)}

4. P = {P1, P2, P3, P4, P5}
P1 = {a,¬b,¬c,¬d}, P2 = {b,¬f}, P3 = {c}, P4 = {d}, P5 = {f}
C = {¬a}

�

{f}

{¬d,¬f}

{¬c,¬d,¬f}

{¬b,¬c,¬d}

{¬a} {a,¬b,¬c,¬d}

{b,¬f}

{c}

{d}

{¬f}

5. Jedno z možných riešeńı:

�

{a}

{a, c}

{a, b} {c,¬b}

{¬c}

{¬a}

{¬e} {¬a, e}

{¬a, d} {e,¬d}

6.
: −a

: −b, c

: −c

�

: −d, c

: −e, c

fail

: −c

�

: −d

: −e

fail

�

7.

: −p(X,X)

: −q(X,Z), r(Z,X)

: −r(b,X)

� [X = a]

: −r(a, b)

fail

: −r(a,X), r(a,X)

fail

: −s(X)

: −t(X, a)

� [X = b]

: −t(X, b)

� [X = a]

: −t(X,X)

fail

8. s(X) :- t(X,a). nahrad́ıme za s(X) :- t(X,a),!.

9 Negácia v logickom programovańı

1. (a) Áno, napŕıklad: S0 obsahuje fakty S1 tretie pravidlo, S2 prvé dve pravidlá.

(b) Nie

(c) Áno, napŕıklad: S0 obsahuje fakty, S1 tretie pravilo, S2 druhé pravidlo, S3 prvé pravidlo.

2. (a) Existuje pre :- ¬ f(5)

(b) Odvodenie neexistuje ani pre jeden z ciel’ov, odvodenia sú blokované. Problémom je, že
program nie je stratifikovaný.

3.
a(P )

c(P ), Y is P + 1, \+ d(Y )

d(P ), Y 1 is P + 1, \+ c(Y 1), Y is P + 1, \d(Y )

[P = 1]⇒ Y 1 is 1 + 1, . . .

\+ c(2), Y is 1 + 1, \+ d(Y )

fail

[P = 2]⇒ Y 1 is 2 + 1, . . .

\+ c(3), Y is 2 + 1, \+ d(Y )

Y is 2 + 1, \+ d(Y )

\+ d(3)

�

c(P ), \+ d(P )

fail vid’ poznámka

c(2)

d(2), Y is 2 + 1, \+ c(Y )

Y is 2 + 1, \+ c(Y )

\+ c(3)

�⇒ \+ c(2) neúspešné

c(3)

d(3), Y is 3 + 1, \+ c(Y )

fail⇒ \+ c(3) úspešné

Poznámka: čast’ stromu bola pre nedostatok miesta vynechaná. V pŕıslušnej vetve ale nikdy
nenastane úspech, ked’že v d’aľsom kroku dostaneme ciel’ v tvare d(P), ..., \+d(P). Vo
vlastnom záujme si však chýbajúcu čast’ stromu nakreslite.

10 Logické programovanie s obmedzujúcimi podmienkami

1. labeling([]).

labeling(Vars) :-

sortByDomainSize(Vars,SortedVars),

SortedVars=[First|Rest],

indomain(First),

label(Rest).

Kde sortByDomainSize je nejaký radiaci algoritmus, ktorý namiesto porovnania X>Y použ́ıva
fd size(X,SizeX), fd size(Y,SizeY), SizeX>SizeY

2. A in {3,5,7,9}, B in 1..4, C in 1..8

3. A in 3..9, B in 1..4, C in 1..8

4. Označme si dané odmedzenia postupne c1 a c2. Dolný index pri hodnote označuje obmedze-
nie, ktoré bolo porušené, pŕıpadne je OK ak bolo nájdené splňujúce priradenie

(a)

1

1c1 2c1 3c1 4c1

2

1

1c2 2c2 3c2 4c2

2c1 3c1 4c1

3

1c1 2

1OK 2c2 3c2 4c2

3c1 4c1

4

1c1 2c1 3

1OK 2OK 3c2 4c2

4c1

(b)

1c1 2⇒c1 B = 1

1c2

3⇒c1 B = 2

2⇒c2 C = 1

1OK

4⇒c1 B = 3

3⇒c2 C in 1..2

1OK 2OK

(c) Pred samostatným prehl’adávańım zaist́ıme hranovú konzistenciu pomocou AC3 - dostávame
A in 3..4, B in 2..3, C in 1..2

3⇒c1 B = 2⇒c2 C = 1

2

1OK

4⇒c1 B = 3

3

1OK 2OK

5. alg([K,C,I,O,A,M,L]) :-

domain([K,I,O,A,L],1,9),

domain([C,M],0,9),

all_distinct([K,C,I,O,A,M,L]),

10*K + C + I #= 10*O + K,

A + A #= 10*K + M,

10*O + L + 10*K + O #= 10*L + I,

10*K + C + A #= 10*O + L,

I + A #= 10*K + O,

10*O + K + 10*K + M #= 10*L + I,

labeling([],[K,C,I,O,A,M,L]).

6. Riešenie bez výpisu:

sudoku4([A1,A2,A3,A4,

B1,B2,B3,B4,

C1,C2,C3,C4,

D1,D2,D3,D4]) :-

Vals = [A1,A2,A3,A4,B1,B2,B3,B4,C1,C2,C3,C4,D1,D2,D3,D4],

domain(Vals,1,4),

all_distinct([A1,A2,A3,A4]),

all_distinct([B1,B2,B3,B4]),

all_distinct([C1,C2,C3,C4]),

all_distinct([D1,D2,D3,D4]),

all_distinct([A1,B1,C1,D1]),

all_distinct([A2,B2,C2,D2]),

all_distinct([A3,B3,C3,D3]),

all_distinct([A4,B4,C4,D4]),

all_distinct([A1,A2,B1,B2]),

all_distinct([A3,A4,B3,B4]),

all_distinct([C1,C2,D1,D2]),

all_distinct([C3,C4,D3,D4]),

labeling([],Vals).

7. Nápoveda: Domy si môžeme oč́ıslovat’ zl’ava doprava č́ıslami 1 až 5. Pre každú národnost’,
farbu, zviera, nápoj a cigarety si vytvoŕıme premennú, ktorá bude nadobúdat’ hodnoty 1 až
5 podl’a pŕıslušného domu. Následne môžeme napŕıklad obmedzenie “človek z prostredného
domu pije mlieko” zaṕısat’ ako Mlieko #= 3 a obmedzenie “Brit býva v červenom dome”
ako Brit #= Cerveny.

Kompletné riešenie:

einstein([Nor,Brit,Sved,Dan,Nemec,

Cerveny,Zeleny,Zlty,Modry,Biely,

Caj,Kava,Mlieko,Pivo,Voda,

Pes,Vtak,Macka,Kon,Ryba,

Pallmall,Dunhill,Blend,Bluemaster,Prince]):-

Vals = [Nor,Brit,Sved,Dan,Nemec,

Cerveny,Zeleny,Zlty,Modry,Biely,

Caj,Kava,Mlieko,Pivo,Voda,

Pes,Vtak,Macka,Kon,Ryba,

Pallmall,Dunhill,Blend,Bluemaster,Prince],

domain(Vals,1,5),

all_distinct([Nor,Brit,Sved,Dan,Nemec]),

all_distinct([Cerveny,Zeleny,Zlty,Modry,Biely]),

all_distinct([Caj,Kava,Mlieko,Pivo,Voda]),

all_distinct([Pes,Vtak,Macka,Kon,Ryba]),

all_distinct([Pallmall,Dunhill,Blend,Bluemaster,Prince]),

%Brit byva v cervenom dome.

Brit #= Cerveny,

%Sved chova psa.

Sved #= Pes,

%Dan pije caj.

Dan #= Caj,

%Zeleny dom stoji hned vlavo od bieleho.

Zeleny #= Biely - 1,

%Majitel zeleneho domu pije kavu.

Zeleny #= Kava,

%Ten, kto fajci Pall Mall, chova vtaka.

Pallmall #= Vtak,

%Majitel zlteho domu fajci Dunhill.

Zlty #= Dunhill,

%clovek z prostredneho domu pije mlieko.

Mlieko #= 3,

%Nor byva v prvom dome.

Nor #= 1,

%Ten, kto fajci Blend, byva vedla toho, kto chova macku.

(Blend #= Macka+1 ; Blend #=Macka-1),

%Ten, kto chova kone, byva vedla toho, kto fajci Dunhill.

(Kon #= Dunhill+1 ; Kon#=Dunhill-1),

%Ten, kto fajci Blue Master, pije pivo.

Bluemaster #= Pivo,

%Nemec fajci Prince.

Nemec #= Prince,

%Nor byva vedla modreho domu.

(Nor #= Modry + 1 ; Nor #= Modry-1 ),

%Ten, kto fajci Blend, ma suseda, ktory pije vodu.

(Blend #= Voda + 1; Blend#= Voda-1),

labeling([],Vals).

11 DC gramatiky

1. Párne č́ısla sú práve tie, ktorých binárny zápis konč́ı nulou. Daná DCG môže vyzerat’

napŕıklad nasledovne:

parne --> [0].

parne --> [1], parne.

parne --> [0], parne.

2. Výsledná gramatika môže vyzerat’ takto:

palindrom --> [].

palindrom --> [a].

palindrom --> [b].

palindrom --> [X],palindrom,[X], {member(X,[a,b])}.

(posledný riadok je možné nahradit’ dvoma rozṕısańım možných hodnôt X)

3. Pre jazyk anbn naṕı̌seme gramatiku

an_bn --> [].

an_bn --> [a],an_bn,[b].

Gramatiku pre jazyk anbm kde m ≥ n potom môžeme zadefinovat’ nasledujúcim spôsobom:

an_bm --> an_bn, same(b).

Pričom same definujeme ako

same(_X) --> [].

same(X) --> [X],same(X).

(teda trochu inak ako bol definovaný na cvičeniach)

4. an_b2n --> [].

an_b2n --> [a],an_bn,[b],[b].

5. Jedno z možných riešeńı je vel’mi podobné riešeniu pre jazyk anbncn, ktoré bolo uvedené
na cvičeniach. Nižšie uvedené riešenie je založené na rovnakej myšlienke (tzn. pomocou
následńıkov), avšak je zaṕısané stručneǰśım spôsobom pomocou pomocného neterminálu
repeat:

abc --> repeat(a,X), repeat(b,s(X)), repeat(c,s(s(X))).

repeat(_,0) --> [].

repeat(A,s(X)) --> [A],repeat(A,X).

6. s(S0,S) :- np(S0,S1), vp(S1,S).

np(S0,S) :- det(S0,S1), n(S1,S).

vp(S0,S) :- tv(S0,S1), np(S1,S).

vp(S0,S) :- v(S0,S).

det([the|S],S).

det([a|S],S).

det([every|S],S).

n([man|S],S).

n([woman|S],S).

n([park|S],S).

tv([loves|S],S).

tv([likes|S],S).

v([walks|S],S).

7. záviśı od vybranej vety

8. S (Ad (Yesterday), S1 (NP ( N (John) ), VP (V (saw), NP (Det (a), NP (Adj (good), NP
(N(movie)))))))
Exp( Sub(Exp( Add(Expr(Num(5)),Exp( Mul(Expr(Num(7)),Exp( Num(2))))))),Exp( Num(4)))

9. sentence(s(Ad,S)) --> adverb(Ad), sentence1(S).

adverb(ad(yesterday)) --> [yesterday].

sentence1(s1(NP,VP)) --> noun_phrase(NP), verb_phrase(VP).

noun_phrase(np(N)) --> noun(N).

noun_phrase(np(Det,NP)) --> determiner(Det), noun_phrase(NP).

noun_phrase(np(Adj,NP)) --> adjective(Adj), noun_phrase(NP).

determiner(det(a)) --> [a].

adjective(adj(good)) --> [good].

noun(n(john)) --> [john].

noun(n(movie)) --> [movie].

verb_phrase(vp(V,NP)) --> verb(V), noun_phrase(NP).

verb(v(saw)) --> [saw].

12 Rôzne

1. foo(X). uspeje práve ak X je prvoč́ıslo. Program môžeme preṕısat’ ako:

prime(X) :- X =\= 1, X1 is X-1, test(X,X1).

test(_X,1) :- !.

test(X,Y) :- X mod Y > 0, Y1 is Y-1, test(X,Y1).

